INTERNATIONAL ATOMIC ENERGY AGENCY
UNITED NATIONS EDUCATIONAL, SCIENTIFIC AND CULTURAL ORGANIZATION
INTERNATIONAL CENTRE FOR THEORETICAL PHYSICS
I.C.T.P., P.O. BOX 586, 34100 TRIESTE, ITALY, CABLE: CENTRATOM TRIESTE

UNITED NATIONS INDUSTRIAL DEVELOPMENT ORGANIZATION
INTERNATIONAL CENTRE FOR SCIENCE AND HIGH TECHNOLOGY

SMR/548-17

Course on Oceanography of Semi-Enclosed Seas
15 April - 3 May 1991

"Fortran & Basic Programs for Computing & Plotting the Astronomic Tide"

F. STRAVISI
University of Trieste
Department of Theoretical Physics
Trieste, Italy

Università degli Studi di Trieste
Facoltà di Scienze

Cattedra di Fisica Terrestre e Climatologia

Franco Stravisi

STANDARD COMPUTATIONS OF PHYSICAL SEAWATER PROPERTIES

Rapporto Interno
FTC 86/3

Trieste 1986

Standard computations of physical seawater properties

Franco Stravisi

ABSTRACT. Fortran subroutines and programs are given for computing the seawater practical salinity, density and some related parameters. The recent international standards proposed by UNESCO are adopted. The software, on magnetic diskettes, can be obtained from the author.

STANDARD SYMBOLS AND UNITS IN OCEANOGRAPHY

The SI system of units is described by UNESCO (1985). Conventions and rules stated in this paper should be used as much as possible. A summary is given here of what may interest physical oceanographers.

Quantity (symbol)	Unit: name	symbol	(relations)
SI base units:			
length	metre	m	
mass	kilogram	kg	
time (t)	second	s	
electric current	ampere	A	
thermodynamic temperature (T)	kelvin	K	
amount of substance	mole	mol	
luminous intensity	candela	cd	
SI derived units with special names:			
frequency	hertz	Hz	(s^{-1})
force	newton	N	($m.kg.s^{-2}$)
pressure (p)	pascal	Pa	(N/m^2)
energy	joule	J	(N.m)
power	watt	W	(J/s)
electric potential	volt	V	(W/A)
electric resistance	ohm	Ω	(V/A)
conductance	siemens	S	(1/Ω)
Celsius temperature (t,θ)	degree Celsius	°C	(K)
SI supplementary units:			
plane angle	radian	rad	(m/m)
solid angle	steradian	sr	(m^2/m^2)
Units in use with SI:			
time (t)	year (tropic)	a	(365.2422d)
	day	d	(24 h)
	hour	h	(60 min)
	minute	min	(60 s)
plane angle	degree	°	(π/180)rad
	minute	'	(1/60)°
	second	"	(1/60)'
pressure (p)	bar	bar	(10^5 Pa)

Oceanographic quantities, symbols and units:

depth (z, positive)	metre	m	
sea pressure (p; p=0 at z=0)	decibar	dbar	(10^4 Pa)
seawater density (ρ)		kg/m^3	
density excess ($\gamma = \rho - 1000\ kg/m^3$)		kg/m^3	
density excess at (p=0) sea level (γ_0)		kg/m^3	
specific volume ($\alpha = 1/\rho$)		m^3/kg	
seawater conductivity (C)		1/(mΩ.cm)	(10^5 S/m)
salinity (practical, S)		psu	(adim.)
chlorinity (Cl)			(adim.)

The kelvin is "the fraction 1/273.16 of the thermodynamic temperature of the triple point of water". The relation between thermodynamic and Celsius temperature is $\theta = T - T_0$, where T_0 = 273.15 K. At present, temperatures are given in the International Practical Temperature Scale 1968 (IPTS-68).

The mole is "the amount of substance of a system which contains as many elementary entities (to be specified as atoms, molecules, ions, electrons or other) as there are atoms in 0.012 kg of carbon-12".

<u>Units to be abandoned</u> are:

litre	1 l	=	1	dm^3
nautical mile			1852	m
knot (1 nautical mile per hour)			0.5144	m/s
conventional millimiter of mercury:	1 mmHg	=	1.333224	hPa
standard atmosphere	1 atm	=	1013.25	hPa
calorie "at 15 °C"	1 cal_{15}	=	4.1855	J
calorie "IT"	1 cal_{IT}	=	4.1868	J
calorie "thermochemical"	1 cal_{th}	=	4.184	J

The term "one standard atmosphere" can be used to indicate the standard pressure at sea level (1013.25 hPa). The (total) pressure at a given depth in the ocean is given by the atmospheric pressure plus the sea pressure.

The adimensional so called "Knudsen's parameter", symbol σ (σ_0 at reference sea level pressure p=0) related to the seawater density <u>relative</u> to the reference pure water density ρ_w = 999.972 kg/m^3 (at 4 °C, 1 atm), is no more used. The relation with the density excess is given by:

$$\gamma = \rho_w (1 + \sigma/1000) - 1000\ kg/m^3 = (0.999972\ \sigma - 0.028)\ kg/m^3 ; \quad (1)$$

with a good approximation in oceanic waters, it is:

$$\gamma = (\sigma - 0.029)\ kg/m^3 . \quad (2)$$

SI prefixes

Power of 10	prefix	symbol	Power of 10	prefix	symbol
12	tera	T	-1	deci	d
9	giga	G	-2	centi	c
6	mega	M	-3	milli	m
3	kilo	k	-6	micro	μ
2	hecto	h	-9	nano	n
1	deca	da	-12	pico	p

The notation rule to state that an oceanographic quantity Q is a function of the three main parameters (practical salinity, temperature and sea pressure) is to write <u>in that order</u> the corresponding simbols or values in parentheses:

$$Q(S,\theta,p), \quad Q(35,15,0), \quad \gamma_t = \gamma(S,\theta,0), \ \ldots \ . \qquad (3)$$

THE PRACTICAL SALINITY SCALE 1978 (PSS-78)

The Practical Salinity Scale 1978 (PSS-78) is defined by UNESCO (1981 a) as follows.

"<u>Absolute salinity</u> (symbol: S_A) is the ratio of mass of dissolved material in seawater to the mass of seawater".

"<u>Practical salinity</u> (symbol: S) is defined in terms of the ratio

$$K_{15}(S) = C(S,15,0)/C_0 \qquad (4)$$

of the electrical conductivity of the seawater sample at 15 °C and 1 atm, to that (C_0) of a potassium chloride (KCl) solution in which the mass fraction of KCl is 0.0324356, at 15 °C and 1 atm, by the following equation:

$$S(x) = 0.0080 - 0.1692\,x + 25.3851\,x^2 + 14.0941\,x^3 - 7.0261\,x^4 + 2.7081\,x^5 , \qquad (5)$$

$$x = \sqrt{K_{15}} .$$

By definition, $K_{15} = 1$ corresponds to S = 35".

The value commonly adopted for the standard conductivity of seawater is:

$$C_0 = C(35,15,0) = 42.909\ (\mathrm{m\Omega.cm})^{-1} = 4.2909\ \mathrm{MS/m} . \qquad (6)$$

Notice that S is an adimensional quantity, three orders of magnitude greater than S_A; the suggested relation, for practical purposes, is:

$$S_A = a\,S + b , \qquad (7)$$

where the coefficients a and b, normally close to 1000 and to 0 respectively, must be determined in the laboratory for any particular oceanic water. Normally, the absolute salinity is "unknown". The present custom, to stress that PSS-78 is adopted, is to write for example "S = 35 psu" instead of "practical salinity S = 35"; the symbol "psu" came therefore into practice to indicate "practical salinity units".

For practical salinity computations some additional parameters are introduced:

$$R\,(S,\theta,p) = C(S,\theta,p)/C_0 \quad \text{(conductivity ratio)}, \qquad (8)$$

$$R_p(S,\theta,p) = C(S,\theta,p)/C(S,\theta,0) , \qquad (9)$$

$$R_t(S,\theta) = C(S,\theta,0)/C(35,\theta,0) , \qquad (10)$$

$$r_t(\theta) = C(35,\theta,0)/C_0 ; \qquad (11)$$

$$R = R_p\, R_t\, r_t . \qquad (12)$$

According to UNESCO (1981 a), practical salinity is computed starting from its definition (5):

$$S = S(K_{15}) = S(R_t(S,15)) = S(R_t(S,\theta)) + DS , \qquad (13)$$

$$DS = (\theta-15)(1+0.0162(\theta-15))^{-1}\, f(\sqrt{R_t}) , \qquad (14)$$

where f(1)=0. Functions f(x), $R_p(R,\theta,p)$ and $r_t(\theta)$ are given in the same reference. Computations are performed by the Fortran function PSS78.

Another parameter of standard use in oceanography is the <u>chlorinity</u> (symbol: Cl), representing the ratio of mass of Cl to the total mass of a seawater sample. The present definition (UNESCO, 1981 a) is: "chlorinity is 0.3285234 times the ratio of the mass of pure reference silver necessary to precipitate the halides contained in the sample of seawater to the mass of this sample". In the hypothesis of constant seawater composition absolute salinity and chlorinity are proportional. According to definitions, chlorinity and practical salinity are independent quantities in describing the properties of seawater; however, since chlorinity is commonly measured in the laboratory by means of the Mohr-Knudsen titration, practical salinity can be directly computed using the accepted ratio:

$$S/\mathrm{psu} = 1806.550\ \mathrm{Cl} . \qquad (15)$$

THE INTERNATIONAL EQUATION OF STATE OF SEAWATER 1980 (IESS-80)

The seawater density $\rho(S,\theta,p)$ is computed as a function of practical salinity, temperature and pressure according to a set of empirical formulas given by UNESCO (1981 b), which constitute the International Equation of State of Seawater 1980 (IESS-80). Practical computations can be performed by means of the Fortran function GAMMA. The principal steps are the following:

$$\rho(S,\theta,p) = \rho(S,\theta,0)/(1-p/K(S,\theta,p)) , \qquad (16)$$

$$K(S,\theta,p) = K(S,\theta,0) + Ap + Bp^2 , \qquad (17)$$

$$\rho(S,\theta,0) = \rho_w(\theta) + \phi(S,\theta) . \qquad (18)$$

The density of the Standard Mean Ocean Water (SMOW) is taken as pure water reference in (18) ($\rho_w(0) = 999.842594\ \mathrm{kg/m^3}$); a future better definition of SMOW in (18) will not change equations (16,17). K is the secant bulk modulus.

ALGORITHMS

Equations and Fortran codes for computing PSS-78, IESS-80 and some other fundamental properties of seawater are described in UNESCO (1983). The corresponding source lists are here reported: some changes have been introduced, mainly for using conductivity as alternate input variable in function PSS78 (UNESCO: SAL78) and to obtain directly the density excess in function GAMMA (UNESCO: SVAN).

Fortran programs, written for using the salinity and density functions, are also given in this report.

FUNCTION PSS78(XI,T,P,M). Performs computations according to PSS-78 (Lewis, 1980). Input XI and output PSS78 depend on the input parameter M as follows:

```
M = 1:     XI = C,   PSS78 = S;
M = 2:     XI = R,   PSS78 = S;
M = 3:     XI = S,   PSS78 = C.
```

Input values are $T=\theta/°C$, $P=p/dbar$; $XI=C,R$ for $M=1,2$; the value (6) has been adopted for the standard seawater conductivity. For $M=3$, the inverse function is yielded. S and C units are psu and $1/(m\Omega.cm)$ respectively.
FUNCTION GAMMA(S,T,P,GAMMAT,SVAN). Computes the seawater density excess (S,θ,p) = GAMMA(S,T,P) kg/m^3 as a function of practical salinity, temperature and sea pressure (Millero et al., 1980; Millero and Poisson, 1981). Output values are also the density excess at atmospheric pressure $\gamma_t(S,\theta)$ = GAMMAT kg/m^3 and the specific volume anomaly $(\alpha(S,\theta,p) - \alpha(35,0,p))$ = SVAN 10^{-8} m^3/kg.
FUNCTION THETA(S,T,P,PR). Computes the potential temperature THETA °C, that is the temperature of a seawater sample adiabatically moved from the in situ (S,T,P) conditions to a reference sea pressure PR, usually that at the sea surface (1 atm, PR = 0).
FUNCTION ATG(S,T,P). Computes the adiabatic temperature gradient (the change of temperature per unit pressure for an adiabatic change of pressure of an element of seawater) ATG °C/dbar at given practical salinity, temperature and sea pressure (Bryden, 1973).
FUNCTION SVEL(S,T,P). Computes the sound velocity SVEL m/s in seawater at given practical salinity, temperature and sea pressure (Chen and Millero, 1977).
FUNCTION DEPTH(P,LAT). Computes pressure (P dbar) to depth (DEPTH m) conversion wit a formula derived from Saunders and Fofonoff (1976); gravity is computed as a function of latitude (LAT degrees) and depth (pressure).
FUNCTION O2(S,T). This additional function computes the fractional volume of oxygen O2 ml/l in seawater at saturation (oxygen solubility x 1000) at standard pressure, at given practical salinity and temperature (Weiss, 1970).

PROGRAM PS78. This Fortran program performs PSS-78 computations through function PSS78 above. Enter parameter M (M=1,2,3 for computing S(C,θ,p), S(R,θ,p), C(S,θ,p) respectively); enter real data on request.
PROGRAM DE80. This Fortran program performs IESS-80 computations through function GAMMA above. Enter real data (S,θ,p): output values are density excess, density excess at standard sea surface pressure, specific volume (times 10^3) and specific volume anomaly (times 10^8).
PROGRAM TPOT. This Fortran programs computes the potential temperature and the adiabatic temperature gradient by using functions THETA and ATG. Input (real) data are practical salinity, temperature, sea pressure and reference pressure.
PROGRAM SOUND. This Fortran programs give the sound speed in seawater by means of function SVEL. Input (real) data are salinity, temperature and pressure.

TABLES

Values of C(S,θ) and of $\gamma_t(S,\theta)$ are reported in Tables 1,2. Values of oxygen solubility are listed in Table 3.

REFERENCES

BRYDEN H.L. (1973): "New polynomials for thermal expansion, adiabatic temperature gradient and potential temperature of seawater", Deep Sea Res., 20, 401-408.

CHEN C.-T., MILLERO F.J. (1977): "Speed of sound in seawater at high pressures", J. Acoust. Soc. of Amer., 62 (5), 1129-1135.

LEWIS E.L. (1980): "The Practical Salinity Scale 1980 and its antecedents", IEEE J.Ocean. Engin., OE-5 (1), 3-8.

MILLERO F.J., CHEN C.-T., BRADSHAW A., SCHLEICHER K. (1980): "A new high pressure equation of state for seawater", Deep Sea Res., 27A, 255-264.

MILLERO F.J., POISSON A. (1981): Summary of data treatment for the UNESCO one atmosphere equation of state for seawater", Deep Sea Res., 28A, 625-629.

SAUNDERS P.M., FOFONOFF N.P. (1976): "Conversion of pressure to depth in the ocean", Deep Sea Res., 23, 109-111.

UNESCO (1981 a): "Background papers and supporting data on the Practical Salinity Scale 1978", Technical Papers Marine Sc., 37, 144 pages.

UNESCO (1981 b): "Background papers and supporting data on the International Equation of State of Seawater 1980", Technical Papers Marine Sc., 38, 192 pages.

UNESCO (1983): "Algorithms for computation of fundamental properties of seawater", Technical Papers Marine Sc., 44, 53 pages.

UNESCO (1985): "The International System of Units (SI) in Oceanography", Technical Papers Marine Sc., 45. Also: IAPSO, Pub. Sc. No. 32.

WEISS R.F. (1970): "The solubility of nitrogen, oxygen and argon in water and seawater", Deep-Sea Res., 17, 721-735.

θ | S	30	31	32	33	34	35	36	37	38	39
1	26.0	26.8	27.6	28.3	29.1	29.9	30.7	31.4	32.2	33.0
2	26.8	27.6	28.4	29.2	30.0	30.8	31.6	32.4	33.1	33.9
3	27.5	28.4	29.2	30.0	30.8	31.7	32.5	33.3	34.1	34.9
4	28.3	29.2	30.0	30.9	31.7	32.5	33.4	34.2	35.0	35.9
5	29.1	30.0	30.9	31.7	32.6	33.5	34.3	35.2	36.0	36.9
6	29.9	30.8	31.7	32.6	33.5	34.4	35.2	36.1	37.0	37.9
7	30.7	31.6	32.5	33.5	34.4	35.3	36.2	37.1	38.0	38.9
8	31.5	32.5	33.4	34.3	35.3	36.2	37.1	38.1	39.0	39.9
9	32.3	33.3	34.3	35.2	36.2	37.1	38.1	39.0	40.0	40.9
10	33.2	34.1	35.1	36.1	37.1	38.1	39.1	40.0	41.0	42.0
11	34.0	35.0	36.0	37.0	38.0	39.0	40.0	41.0	42.0	43.0
12	34.8	35.9	36.9	37.9	39.0	40.0	41.0	42.0	43.0	44.1
13	35.7	36.7	37.8	38.9	39.9	41.0	42.0	43.0	44.1	45.1
14	36.5	37.6	38.7	39.8	40.9	41.9	43.0	44.1	45.1	46.2
15	37.4	38.5	39.6	40.7	41.8	42.9	44.0	45.1	46.2	47.3
16	38.2	39.4	40.5	41.6	42.8	43.9	45.0	46.1	47.2	48.3
17	39.1	40.3	41.4	42.6	43.7	44.9	46.0	47.2	48.3	49.4
18	40.0	41.2	42.4	43.5	44.7	45.9	47.1	48.2	49.4	50.5
19	40.9	42.1	43.3	44.5	45.7	46.9	48.1	49.3	50.5	51.6
20	41.7	43.0	44.2	45.5	46.7	47.9	49.1	50.3	51.5	52.7
21	42.6	43.9	45.2	46.4	47.7	48.9	50.2	51.4	52.6	53.9
22	43.5	44.8	46.1	47.4	48.7	50.0	51.2	52.5	53.7	55.0
23	44.4	45.7	47.1	48.4	49.7	51.0	52.3	53.6	54.8	56.1
24	45.3	46.7	48.0	49.4	50.7	52.0	53.3	54.6	56.0	57.3
25	46.2	47.6	49.0	50.4	51.7	53.1	54.4	55.7	57.1	58.4
26	47.2	48.6	50.0	51.3	52.7	54.1	55.5	56.8	58.2	59.5
27	48.1	49.5	50.9	52.3	53.8	55.2	56.6	57.9	59.3	60.7
28	49.0	50.5	51.9	53.3	54.8	56.2	57.6	59.1	60.5	61.9
29	49.9	51.4	52.9	54.4	55.8	57.3	58.7	60.2	61.6	63.0
30	50.9	52.4	53.9	55.4	56.9	58.3	59.8	61.3	62.7	64.2

Tab. 1. Seawater conductivity /(mΩ.cm)⁻¹ at given salinity (30-39 psu) and temperature (1-30 °C).

S | θ	8	10	12	14	16	18	20	22	24	26
20.0	15.5	15.3	15.0	14.6	14.2	13.8	13.4	12.9	12.3	11.8
20.5	15.9	15.7	15.4	15.0	14.6	14.2	13.7	13.2	12.7	12.1
21.0	16.3	16.0	15.7	15.4	15.0	14.6	14.1	13.6	13.1	12.5
21.5	16.7	16.4	16.1	15.8	15.4	15.0	14.5	14.0	13.5	12.9
22.0	17.1	16.8	16.5	16.2	15.8	15.3	14.9	14.4	13.8	13.3
22.5	17.5	17.2	16.9	16.6	16.2	15.7	15.3	14.8	14.2	13.6
23.0	17.9	17.6	17.3	16.9	16.5	16.1	15.6	15.1	14.6	14.0
23.5	18.3	18.0	17.7	17.3	16.9	16.5	16.0	15.5	15.0	14.4
24.0	18.7	18.4	18.1	17.7	17.3	16.9	16.4	15.9	15.3	14.8
24.5	19.0	18.8	18.5	18.1	17.7	17.3	16.8	16.3	15.7	15.1
25.0	19.4	19.2	18.8	18.5	18.1	17.6	17.2	16.6	16.1	15.5
25.5	19.8	19.5	19.2	18.9	18.5	18.0	17.5	17.0	16.5	15.9
26.0	20.2	19.9	19.6	19.2	18.8	18.4	17.9	17.4	16.8	16.3
26.5	20.6	20.3	20.0	19.6	19.2	18.8	18.3	17.8	17.2	16.6
27.0	21.0	20.7	20.4	20.0	19.6	19.2	18.7	18.2	17.6	17.0
27.5	21.4	21.1	20.8	20.4	20.0	19.5	19.1	18.5	18.0	17.4
28.0	21.8	21.5	21.2	20.8	20.4	19.9	19.4	18.9	18.4	17.8
28.5	22.2	21.9	21.5	21.2	20.8	20.3	19.8	19.3	18.7	18.1
29.0	22.6	22.3	21.9	21.6	21.1	20.7	20.2	19.7	19.1	18.5
29.5	23.0	22.7	22.3	21.9	21.5	21.1	20.6	20.0	19.5	18.9
30.0	23.4	23.1	22.7	22.3	21.9	21.4	21.0	20.4	19.9	19.3
30.5	23.7	23.4	23.1	22.7	22.3	21.8	21.3	20.8	20.2	19.6
31.0	24.1	23.8	23.5	23.1	22.7	22.2	21.7	21.2	20.6	20.0
31.5	24.5	24.2	23.9	23.5	23.1	22.6	22.1	21.6	21.0	20.4
32.0	24.9	24.6	24.3	23.9	23.4	23.0	22.5	21.9	21.4	20.8
32.5	25.3	25.0	24.6	24.3	23.8	23.4	22.9	22.3	21.8	21.2
33.0	25.7	25.4	25.0	24.6	24.2	23.7	23.2	22.7	22.1	21.5
33.5	26.1	25.8	25.4	25.0	24.6	24.1	23.6	23.1	22.5	21.9
34.0	26.5	26.2	25.8	25.4	25.0	24.5	24.0	23.5	22.9	22.3
34.5	26.9	26.6	26.2	25.8	25.4	24.9	24.4	23.8	23.3	22.7
35.0	27.3	27.0	26.6	26.2	25.7	25.3	24.8	24.2	23.6	23.0
35.5	27.7	27.3	27.0	26.6	26.1	25.7	25.1	24.6	24.0	23.4
36.0	28.1	27.7	27.4	27.0	26.5	26.0	25.5	25.0	24.4	23.8
36.5	28.5	28.1	27.8	27.3	26.9	26.4	25.9	25.4	24.8	24.2
37.0	28.8	28.5	28.1	27.7	27.3	26.8	26.3	25.7	25.2	24.5
37.5	29.2	28.9	28.5	28.1	27.7	27.2	26.7	26.1	25.5	24.9
38.0	29.6	29.3	28.9	28.5	28.1	27.6	27.1	26.5	25.9	25.3
38.5	30.0	29.7	29.3	28.9	28.4	28.0	27.4	26.9	26.3	25.7
39.0	30.4	30.1	29.7	29.3	28.8	28.3	27.8	27.3	26.7	26.1
39.5	30.8	30.5	30.1	29.7	29.2	28.7	28.2	27.6	27.1	26.4
40.0	31.2	30.9	30.5	30.1	29.6	29.1	28.6	28.0	27.4	26.8

Tab. 2. Seawater density excess /$kg.m^{-3}$ at standard sea surface pressure at given temperature (8-26 °C) and salinity (20-40 psu).

θ | S	30	31	32	33	34	35	36	37	38	39
1	8.11	8.05	8.00	**7.94**	7.89	**7.84**	7.78	**7.73**	7.68	**7.63**
2	7.90	7.85	7.79	**7.74**	7.69	**7.64**	7.59	**7.53**	7.48	**7.43**
3	7.70	7.65	7.60	**7.55**	7.50	**7.45**	7.40	**7.35**	7.30	**7.25**
4	7.51	7.46	7.41	**7.36**	7.31	**7.26**	7.22	**7.17**	7.12	**7.07**
5	7.33	7.28	7.23	**7.18**	7.14	**7.09**	7.04	**7.00**	6.95	**6.90**
6	7.15	7.11	7.06	**7.01**	6.97	**6.92**	6.88	**6.83**	6.79	**6.74**
7	6.98	6.94	6.89	**6.85**	6.81	**6.76**	6.72	**6.67**	6.63	**6.59**
8	6.82	6.78	6.74	**6.69**	6.65	**6.61**	6.57	**6.52**	6.48	**6.44**
9	6.67	6.63	6.59	**6.54**	6.50	**6.46**	6.42	**6.38**	6.34	**6.30**
10	6.52	6.48	6.44	**6.40**	6.36	**6.32**	6.28	**6.24**	6.20	**6.16**
11	6.38	6.34	6.30	**6.26**	6.22	**6.18**	6.14	**6.10**	6.07	**6.03**
12	6.24	6.20	6.17	**6.13**	6.09	**6.05**	6.01	**5.98**	5.94	**5.90**
13	6.11	6.07	6.04	**6.00**	5.96	**5.93**	5.89	**5.85**	5.82	**5.78**
14	5.99	5.95	5.91	**5.88**	5.84	**5.80**	5.77	**5.73**	5.70	**5.66**
15	5.87	5.83	5.79	**5.76**	5.72	**5.69**	5.65	**5.62**	5.58	**5.55**
16	5.75	5.71	5.68	**5.64**	5.61	**5.58**	5.54	**5.51**	5.48	**5.44**
17	5.64	5.60	5.57	**5.53**	5.50	**5.47**	5.43	**5.40**	5.37	**5.34**
18	5.53	5.49	5.46	**5.43**	5.40	**5.36**	5.33	**5.30**	5.27	**5.24**
19	5.42	5.39	5.36	**5.33**	5.29	**5.26**	5.23	**5.20**	5.17	**5.14**
20	5.32	5.29	5.26	**5.23**	5.20	**5.17**	5.14	**5.10**	5.07	**5.05**
21	5.22	5.19	5.16	**5.13**	5.10	**5.07**	5.04	**5.01**	4.98	**4.95**
22	5.13	5.10	5.07	**5.04**	5.01	**4.98**	4.95	**4.92**	4.89	**4.87**
23	5.04	5.01	4.98	**4.95**	4.92	**4.89**	4.87	**4.84**	4.81	**4.78**
24	4.95	4.92	4.89	**4.86**	4.84	**4.81**	4.78	**4.75**	4.73	**4.70**
25	4.86	4.84	4.81	**4.78**	4.75	**4.73**	4.70	**4.67**	4.65	**4.62**
26	4.78	4.75	4.73	**4.70**	4.67	**4.65**	4.62	**4.59**	4.57	**4.54**
27	4.70	4.67	4.65	**4.62**	4.60	**4.57**	4.54	**4.52**	4.49	**4.47**
28	4.62	4.60	4.57	**4.55**	4.52	**4.60**	4.47	**4.45**	4.42	**4.40**
29	4.55	4.52	4.50	**4.47**	4.45	**4.42**	4.40	**4.37**	4.35	**4.33**
30	4.47	4.45	4.43	**4.40**	4.38	**4.35**	4.33	**4.31**	4.28	**4.26**

Tab. 3. Oxigen solubility /(ml/l) in seawater at given salinity (30-39 psu) and temperature (1-30 °C).


```
      FUNCTION PSS78(XI,T,P,M)
C-----------------------------------------------Franco Stravisi 1986
C  Practical Salinity Scale 1978 (PSS-78),
C  Unesco TP 37, 1981.
C  Code from: Unesco TP 44, 1983, pages 11-12 (SAL78).
C  T    seawater temperature /°C (IPTS 1968),
C  P    sea pressure /dbar (P=0 at z=0),
C  S    practical salinity /psu (PSS-78),
C  C    conductivity  C(S,T,P) /(1/(mohm.cm)).
C
C  Definitions according to UNESCO TP 37:
C  CO = C(35,15,0) ;           here: CO = 42.909  1/(mohm.cm);
C  R   = C(S,T,P)/CO           conductivity ratio in situ;
C  RT = C(S,T,0)/C(35,T,0)     conductivity ratio at P=0;
C  RMT= C(35,T,0)/CO              (5)
C  RP = C(S,T,P)/C(S,T,0)         (4)
C  RT = R/(RPxRMT)                (3)
C  S  = S(RT,T)                   (6)
C
C  INPUT/OUTPUT according to code M:
C  M      =       1      2      3
C  XI     =       C      R      S
C  PSS78 =        S      S      C
C
C  NB:   PSS78 = 0      for  R < 0.0005  (M=1,2)
C                   and for  S < 0.02   (M=3).
C-------------------------------------------------------------------
C--- S(RT,T)   (TP 37/6)       (XT=T-15; XR=SQRT(RT)) :
      SAL(XR,XT)=
     .   (((((2.7081*XR-7.0261)*XR+14.0941)*XR+25.3851)*XR
     .   -0.1692)*XR+0.0080+(XT/(1.0+0.0162*XT))*
     .   (((((-0.0144*XR+0.0636)*XR-0.0375)*XR-0.0066)
     .   *XR-0.0056)*XR+0.0005)
C--- Derivative of S with respect to XR :
      DSAL(XR,XT)=((((13.5405*XR-28.1044)*XR+42.2823)
     .   *XR+50.7702)*XR.-0.1692)+(XT/(1.0+0.0162*XT))*
     .   ((((-0.0720*XR+0.2544)*XR-0.1125)*XR-0.0132)
     .   *XR-0.0056)
C--- RMT(T)    (TP 37/5)  :
      RMT(XT)=(((1.0031E-9*XT-6.9698E-7)*XT+1.104259E-4)*XT
     .        +2.00564E-2)*XT+0.6766097
C--- Coefficients of   RP(R,T,P), for P in dbar (TP 37/4) :
      E123(XP)=((3.989E-15*XP-6.370E-10)*XP+2.070E-5)*XP
      D12(XT)=(4.464E-4*XT+3.426E-2)*XT+1.0
      D34(XT)=-3.107E-3*XT+0.4215
C---
      XX=XI
C--- Translate  C in  R = C/CO  :
      IF(M.EQ.1) XX=XX/42.909
C--- Yield zero for small XX:
      PSS78=0.0
      IF((M.LE.2).AND.(XX.LE.5E-4)) RETURN
      IF((M.EQ.3).AND.(XX.LE.0.02)) RETURN
C---
      DT=T-15.0
C---  Select direct (M=1/2) or inverse (M=3) function:
```

```
      IF(M.EQ.3) GOTO 10
C---  From  R (or C)  to S :
      R=XX
      RT=R/(RMT(T)*(1.0+E123(P)/(D12(T)+D34(T)*R)))
      RT=SQRT(ABS(RT))
      PSS78=SAL(RT,DT)
      RETURN
C---  From  S  to  C ; inversion by means of the
C---  Newton-Raphson iteration (max 10 cicles) :
   10 RT=SQRT(XX/35.0)
      SI=SAL(RT,DT)
      N=0
   15 RT=RT+(XX-SI)/DSAL(RT,DT)
      SI=SAL(RT,DT)
      N=N+1
      DELS=ABS(SI-XX)
      IF((DELS.GT.1.0E-4).AND.(N.LT.10)) GOTO 15
C---
      RTT=RMT(T)*RT*RT
      AT=D34(T)
      BT=D12(T)
      CP=E123(P)
      CP=RTT*(CP+BT)
      BT=BT-RTT*AT
      R=SQRT(ABS(BT*BT+4.0*AT*CP))-BT
      PSS78=42.909*0.5*R/AT
      RETURN
      END
```

```
      FUNCTION GAMMA(S,T,P,GAMMAT,SVAN)
C------------------------------------------ Franco Stravisi 1986
C  Computes the density excess GAMMA = 1/V(S,t,p)-1000 kg/m3
C  and the specific volume anomaly SVAN = V(S,t,p)-V(35,0,p)
C  of seawater by means of IESS 1980, UNESCO TP 38, 1981.
C  From: UNESCO TP 44, 1983, 20-21.
C  INPUT:
C  S        practical salinity /psu (PSS-78),
C  T        temperature /°C (IPTS-68),
C  P        sea pressure /dbar.
C  OUTPUT:
C  GAMMA    density excess /(kg/m3),
C  GAMMAT   density excess /(kg/m3) at P = 0 ,
C  SVAN     specific volume anomaly /(1E-8 m3/kg).
C
C  NB :     seawater density /(kg/m3) is  1000 + GAMMA.
C---------------------------------------------------------------
      REAL K,K0,KW,K35
      EQUIVALENCE (E,D,B1),(BW,B,R3),(C,A1,R2)
      EQUIVALENCE (AW,A,R1),(KW,K0,K)
      DATA R3500,R4/1028.1063,4.8314E-4/
      DATA DR350/28.106331/
      PB=P/10.
```



```
      SR=SQRT(ABS(S))
C-- One Atmosphere International Equation of State of
C   Seawater, 1980 --
C-- Pure water density (P=0) :
      R1=((((6.536332E-9*T-1.120083E-6)*T+1.001685E-4)*T
     .-9.095290E-3)*T+6.793952E-2)*T-28.263737
C-- Seawater density (P=0) :
      R2=(((5.3875E-9*T-8.2467E-7)*T+7.6438E-5)*T
     .  -4.0899E-3)*T+8.24493E-1
      R3=(-1.6546E-6*T+1.0227E-4)*T-5.72466E-3
C-- IESS80 (P=0)
      SIG=(R4*S+R3*SR+R2)*S+R1
C-- Specific volume (P=0) :
      V350P=1.0/R3500
      SVA=-SIG*V350P/(R3500+SIG)
      GAMMAT=SIG+DR350
      GAMMA=GAMMAT
      SVAN=SVA*1.0E8
      IF(PB.EQ.0.0) RETURN
C-- High Pressure International Equation of State of
C   Seawater, 1980 --
      E=(9.1697E-10*T+2.0816E-8)*T-9.9348E-7
      BW=(5.2787E-8*T-6.12293E-6)*T+3.47718E-5
      B=BW+E*S
      D=1.91075E-4
      C=(-1.6078E-6*T-1.0981E-5)*T+2.2838E-3
      AW=((-5.77905E-7*T+1.16092E-4)*T+1.43713E-3)*T
     .  -0.1194975
      A=(D*SR+C)*S+AW
      B1=(-5.3009E-4*T+1.6483E-2)*T+7.944E-2
      A1=((-6.1670E-5*T+1.09987E-2)*T-0.603459)*T+54.6746
      KW=(((-5.155288E-5*T+1.360477E-2)*T-2.327105)*T
     .   +148.4206)*T-1930.06
      K0=(B1*SR+A1)*S+KW
      DK=(B*PB+A)*PB+K0
      K35=(5.03217E-5*PB+3.359406)*PB+21582.27
      GAM=PB/K35
      PK=1.0-GAM
      SVA=SVA*PK+(V350P+SVA)*PB*DK/(K35*(K35+DK))
      SVAN=SVA*1.0E+8
      V350P=V350P*PK
      DR35P=GAM/V350P
      DVAN=SVA/(V350P*(V350P+SVA))
      GAMMA=DR350+DR35P-DVAN
      RETURN
      END
```

```
      FUNCTION THETA(S,T,P,PR)
C------------------------------------- Franco Stravisi 1986
C   Computes potential temperature THETA /°C of seawater:
C   S      practical salinity /psu  (PSS-1978),
C   T      temperature /°C,
C   P      sea pressure /dbar,
C   PR     reference pressure /dbar.
C   From: UNESCO TP 44, 1983, 44.
C------------------------------------------------------------
      P0=P
      T0=T
      H=PR-P0
      XK=H*ATG(S,T0,P0)
      T0=T0+0.5*XK
      Q=XK
      P0=P0+0.5*H
      XK=H*ATG(S,T0,P0)
      T0=T0+0.29289322*(XK-Q)
      Q=0.58578644*XK+0.121320344*Q
      XK=H*ATG(S,T0,P0)
      T0=T0+1.707106781*(XK-Q)
      Q=3.414213562*XK-4.121320344*Q
      P0=P0+0.5*H
      XK=H*ATG(S,T0,P0)
      THETA=T0+(XK-2.0*Q)/6.0
      RETURN
      END
```

```
      FUNCTION ATG(S,T,P)
C------------------------------------- Franco Stravisi 1986
C   Computes the adiabatic temperature gradient.
C   From: UNESCO TP 44, 1983, 40.
C   S     practical salinity /psu (PSS-1978),
C   T     seawater temperature /°C,
C   P     sea pressure /dbar,
C   ATG   adiabatic temperature gradient (°C/dbar).
C------------------------------------------------------------
      DS=S-35.0
      ATG=(((-2.1687E-16*T+1.8676E-14)*T-4.6206E-13)*P
     .   +((2.7759E-12*T-1.1351E-10)*DS+((-5.4481E-14*T
     .   +8.733E-12)*T-6.7795E-10)*T+1.8741E-8))*P
     .   +(-4.2393E-8*T+1.8932E-6)*DS
     .   +((6.6228E-10*T-6.836E-8)*T+8.5258E-6)*T+3.5803E-5
      RETURN
      END
```



```
      FUNCTION SVEL(S,T,P)
C------------------------------------- Franco Stravisi 1986
C   Computes sound velocity in seawater.
C   From: UNESCO TP 44, 1983, 49.
C   S     practical salinity /psu (PSS-1978),
C   T     temperature /°C (IPTS-1968),
C   P     sea pressure /dbar,
C   SVEL  sound velocity /(m/s).
C------------------------------------------------------------
      EQUIVALENCE (A0,B0,C0),(A1,B1,C1),(A2,C2),(A3,C3)
      PB=P/10.
      SR=SQRT(ABS(S))
      D=1.727E-3 -7.9836E-6*PB
      B1=7.3637E-5 +1.7945E-7*T
      B0=-1.922E-2 -4.42E-5*T
      B=B0+B1*PB
      A3=(-3.389E-13*T+6.649E-12)*T+1.100E-10
      A2=((7.988E-12*T-1.6002E-10)*T+9.1041E-9)*T-3.9064E-7
      A1=(((-2.0122E-10*T+1.0507E-8)*T-6.4885E-8)*T
     .  -1.2580E-5)*T+9.4742E-5
      A0=(((-3.21E-8*T+2.006E-6)*T+7.164E-5)*T-1.262E-2)*T
     .  +1.389
      A=((A3*PB+A2)*PB+A1)*PB+A0
      C3=(-2.3643E-12*T+3.8504E-10)*T-9.7729E-9
      C2=(((1.0405E-12*T-2.5335E-10)*T+2.5974E-8)*T
     .  -1.7107E-6)*T+3.1260E-5
      C1=(((-6.1185E-10*T+1.3621E-7)*T-8.1788E-6)*T
     .  +6.8982E-4)*T+0.153563
      C0=((((3.1464E-9*T-1.47800E-6)*T+3.3420E-4)*T
     .  -5.80852E-2)*T+5.03711)*T+1402.388
      C=((C3*PB+C2)*PB+C1)*PB+C0
      SVEL=C+(A+B*SR+D*S)*S
      RETURN
      END
```

```
      FUNCTION DEPTH(P,LAT)
C----------------------------------- Franco Stravisi 1986
C   Computes DEPTH /m for a standard ocean
C   (T = 0 °C,  S = 35 psu), at given:
C   P       sea pressure /dbar,
C   LAT     latitude /degrees
C   From: UNESCO TP 44, 1983, 28.
C----------------------------------------------------------
      REAL LAT
      X=SIN(LAT/57.29578)
      X=X*X
      GR=9.780318*(1.0+(5.2788E-3+2.36E-5*X)*X)+1.092E-6*P
      DEPTH=((((-1.82E-15*P+2.279E-10)*P-2.2512E-5)*P
     .      +9.72659)*P)/GR
      RETURN
      END
```

```
      FUNCTION O2(S,TC)
C------------------------------------------------ Franco Stravisi 1980
C  Computes O2 solubility /(ml/l) in seawater.
C  Reference: Weiss (1970), Deep Sea Res., 17, 721-735.
C  S  practical salinity /psu , TC temperature /°C .
C---------------------------------------------------------------------
      T=(TC+273.15)/100.
      CL=-173.4292+249.6339/T+143.3483*ALOG(T)-21.8492*T+
     .   S*(-0.033096+0.014259*T-0.0017000*T*T)
      O2=EXP(CL)
      RETURN
      END
```

```
      PROGRAM PS78
C---------------------------------------------------------------------
C  Computes practical salinity/conductivity.
C  Function required: PSS78
C---------------------------------------------------------------------
      WRITE(*,1) ' --- Practical Salinity Scale 1978 ------'
      WRITE(*,1) ' --- (Franco Stravisi 1986) -------------'
      WRITE(*,1) ' C(S,t,p)    conductivity (1/(mohm.cm)), '
      WRITE(*,1) ' R(S,t,p)    conductivity ratio,         '
      WRITE(*,1) ' S           practical salinity;         '
      WRITE(*,1) '             (S/psu, t/°C, p/dbar).'
      WRITE(*,1) ' ----------------------------------------'
      WRITE(*,1) ' M = 1       C  ---> S                   '
      WRITE(*,1) ' M = 2       R  ---> S                   '
      WRITE(*,1) ' M = 3       S  ---> C                   '
      WRITE(*,1) ' ----------------------------------------'
      WRITE(*,1) ' Enter parameter M. Enter REAL data.'
      WRITE(*,1) ' ----------------------------------------'
      WRITE(*,2) ' M  ?   '
    1 FORMAT(A43)
    2 FORMAT(A\)
      READ(*,3) M
    3 FORMAT(I5)
  100 CONTINUE
      IF(M.EQ.1) WRITE(*,2) ' C  ?    '
      IF(M.EQ.2) WRITE(*,2) ' R  ?    '
      IF(M.EQ.3) WRITE(*,2) ' S  ?    '
      READ(*,4) XX
      WRITE(*,2) ' t  ?   '
      READ(*,4) T
      WRITE(*,2) ' p  ?   '
      READ(*,4) P
    4 FORMAT(F15.7)
      X=PSS78(XX,T,P,M)
      IF(M.EQ.1) WRITE(*,5) X
      IF(M.EQ.2) WRITE(*,5) X
      IF(M.EQ.3) WRITE(*,6) X
    5 FORMAT(' S  =    ',F9.3/)
    6 FORMAT(' C  =    ',F9.3/)
      GOTO 100
      STOP
      END
```



```
      PROGRAM DE80
C---------------------------------------------------------------------
C  Computes the seawater density (IESS 1980) at
C  (S,t,p) and at (S,t,0) and the specific volume anomaly.
C  Function required: GAMMA.
C---------------------------------------------------------------------
      WRITE(*,5) ' ---- (Franco Stravisi 1986) -----------'
      WRITE(*,5) ' Computes gamma, gamma-t and specific   '
      WRITE(*,5) ' volume of seawater (IESS-1980).        '
      WRITE(*,5) ' INPUT :                                '
      WRITE(*,5) ' S    salinity /psu (PSS-1978),         '
      WRITE(*,5) ' t    temperature /°C,                  '
      WRITE(*,5) ' p    sea pressure /dbar).              '
      WRITE(*,5) ' OUTPUT :                               '
      WRITE(*,5) ' density excess at (S,T,P) /(kg/m3),    '
      WRITE(*,5) ' density excess at (S,T,0) /(kg/m3),    '
      WRITE(*,5) ' specific volume /(m3/kg) x 10**3.      '
      WRITE(*,5) ' specific volume anomaly  x 10**8.      '
      WRITE(*,5) ' ---------------------------------------'
      WRITE(*,5) ' Enter REAL data.                       '
      WRITE(*,5) ' ---------------------------------------'
    5 FORMAT(A44)
  100 WRITE(*,1) ' S ? '
    1 FORMAT(A\)
      READ(*,2) S
    2 FORMAT(F15.7)
      WRITE(*,1) ' t ? '
      READ(*,2) T
      WRITE(*,1) ' p ? '
      READ(*,2) P
      GA=GAMMA(S,T,P,GT,SV)
      VOL=1000./(1000.+GA)
      WRITE(*,3) GA,GT,VOL,SV
    3 FORMAT(' gamma    =   ',F8.4,'        kg/m3'/
     .       ' gamma-t  =   ',F8.4,'        kg/m3'/
     .       ' sp. vol. =    ',F8.5,'  E-3  m3/kg'/
     .       ' sv anom.=',F8.2,'      E-8  m3/kg'/)
      GOTO 100
      STOP
      END
```

```
      PROGRAM TPOT
C-------------------------------------------------------------
C   Computes the potential temperature and the adiabatic
C   temperature gradient.
C   Functions required: THETA and ATG.
C-------------------------------------------------------------
      WRITE(*,3) ' --- (Franco Stravisi 1986) -------------'
      WRITE(*,3) ' Computes potential temperature /°C and  '
      WRITE(*,3) ' adiabatic temperature gradient/(°C/dbar)'
      WRITE(*,3) ' ----------------------------------------'
    3 FORMAT(A41)
  100 WRITE(*,1) ' Salinity /psu        ?    '
    1 FORMAT(A\)
      READ(*,2) S
    2 FORMAT(F15.5)
      WRITE(*,1) ' Temperature /°C      ?    '
      READ(*,2) T
      WRITE(*,1) ' Pressure     /dbar   ?    '
      READ(*,2) P
      WRITE(*,1) ' Ref. press. /dbar    ?    '
      READ(*,2) PR
      GR=1000.*ATG(S,T,P)
      TH=THETA(S,T,P,PR)
      WRITE(*,5) GR,TH
    5 FORMAT(' adiab. t gradient  =',F8.4,' E-3 °C/dbar'/
     .       ' potential temper.  =',F7.3,'   °C'/)
      GOTO 100
      END
```

```
      PROGRAM SOUND
C-------------------------------------------------
C   Computes sound velocity in seawater.
C   Function required: SVEL.
C-------------------------------------------------
      WRITE(*,1) ' --- (Franco Stravisi 1986) -----'
      WRITE(*,1) ' Computes sound velocity C(S,T,P)'
      WRITE(*,1) ' in seawater /(m/s).             '
      WRITE(*,1) ' --------------------------------'
    1 FORMAT(A33)
   10 WRITE(*,2) ' Salinity     /psu  ?    '
    2 FORMAT(A\)
      READ(*,3) S
    3 FORMAT(F10.5)
      WRITE(*,2) ' Temperature /°C    ?    '
      READ(*,3) T
      WRITE(*,2) ' Pressure     /dbar ?    '
      READ(*,3) P
      C=SVEL(S,T,P)
      WRITE(*,5) C
    5 FORMAT(' Sound velocity         = ',F10.1,' m/s'/)
      GOTO 10
      END
```